CIIC 4030/ICOM 4036 Programming Languages
Assignment #2

Given the lexical analyzer implemented in assignment #1, add the parsing code using PLY to recognize the following context free grammar:

```
defdefs -> defdef defdefs
defdefs -> defdef
defdef -> DEF ID LPAREN parmsopt RPAREN COLON type BECOMES LBRACE vardefsopt
defdefsopt expras RBRACE
parmsopt -> parms
parmsopt ->
parms -> vardef COMMA parms
parms -> vardef
vardef -> ID COLON type
type -> INT
type -> LPAREN typesopt RPAREN ARROW type
typesopt -> types
typesopt ->
types -> type COMMA types
types -> type
vardefsopt -> VAR vardef SEMI vardefsopt
vardefsopt ->
defdefsopt -> defdefs
defdefsopt ->
expras -> expra SEMI expras
expras -> expra
expra -> ID BECOMES expr
expra -> expr
expr -> IF LPAREN test RPAREN LBRACE expras RBRACE ELSE LBRACE expras RBRACE
expr -> term
expr -> expr PLUS term
expr -> expr MINUS term
term -> factor
term -> term STAR factor
term -> term SLASH factor
term -> term PCT factor
factor -> ID
factor -> NUM
factor -> LPAREN expr RPAREN
factor -> factor LPAREN argsopt RPAREN
test -> expr NE expr
test -> expr LT expr
test -> expr LE expr
test -> expr GE expr
test -> expr GT expr
test -> expr EQ expr
argsopt -> args
argsopt ->
args -> expr COMMA args
args -> expr
```

Note: Your program must be tested with two files provided. Write a short report (1 page) describing the results of each test and the experience developing the scanner/parser using PLY. Submit both code and short report in Moodle.